

Example Using SPSS Matrix Procedure

```
matrix.
***** define a matrix with the compute command*****.

***** A is a 2 X 2 matrix*****.
compute A={3,2;4,0}.

***** B is a 2 X 5 matrix*****.
compute B={1,3,0,5,7;2,4,3,3,3}.

***** C is a 2 X 2 matrix*****.
compute C={0,8;9,1}.

***** transpose B*****.
compute Btrans=transpos(B).

***** Add *****.
compute AplusC=A+C.

***** multiply*****.
compute AB=A*B.

***** inverse*****.
compute inverseA=inv(A).

***** determinant ***.
compute detA=det(A).

*****print the results *****.
print A /title "Matrix A".
print B /title "Matrix B".
print Btrans /title "B transpose".
print AplusC /title "A + C".
print AB /title "AB".
print inverseA /title "A inverse".
print detA /title "Det A".

end matrix.
```

Results

Matrix

Run MATRIX procedure:

Matrix A

```
3 2
4 0
```

Matrix B

```
1 3 0 5 7
2 4 3 3 3
```

B transpose

```
1 2
3 4
0 3
5 3
7 3
```

A + C

```
3 10
13 1
```

AB

```
7 17 6 21 27
4 12 0 20 28
```

A inverse

```
.0000000000 .2500000000
.5000000000 -.3750000000
```

Det A

```
-8
```

----- END MATRIX -----

Computing a Covariance Matrix using Matrix Algebra

To compute a covariance matrix with SPSS, follow the formula below.

$$\text{cov}(X, Y) = \left(\frac{1}{n-1} \right) X'X$$

where the matrix above called X is a matrix of X and Y variables with (one column each). In SPSS, then, you would want to specify just one matrix that contains both the X and Y variables.

Computing the Correlation Matrix using Matrix Algebra

$$R = D^{-1/2} S D^{-1/2}$$

Where D is the diagonal matrix of variances. In other words, $D^{-1/2}$ is a diagonal matrix with $1/\text{sd}$ for each variable as the diagonal elements.

Computing the Regression Coefficient using Matrix Algebra

To obtain the regression coefficient, follow the formula below.

$$b = (X'X)^{-1} X'Y$$

where Y is a column vector ($n \times 1$) and X is an $n \times k$ matrix with the first column as ones and as many additional columns as variables and as many rows as cases ($n \times k + 1$). By using ones in the first column of the matrix, you obtain the intercept as the first coefficient in the resulting matrix. Contrary to the covariance example above where X and Y were in the same matrix, the regression coefficient formula requires that X and Y have separate matrices.

Computing Basic Statistical Values with SPSS Matrix Procedure

```

**** This file computes some basic statistical values using matrix algebra.
matrix.
*compute the mean of X1 and X2.
compute X={-2,-1;
           0,1;
           2,3;
           4,-2;
           -4,-1}.
compute ONE=make(5,1,1). /* (5,1,1) generates a matrix with 5 rows, 1 column, and elements equal to 1.
compute transONE=transpos(ONE).
compute meanX=transONE*X*(1/5).
print meanX /title "mean of X1 and X2"
           /format="F5.2".

**** This section computes a variance-covariance matrix, s, for two variables, X1 and X2.
compute X={-2,-1;
           0,1;
           2,3;
           4,-2;
           -4,-1}.

compute Xtrans=transpos(X).
compute Xprod=Xtrans*X.
compute S=(1/(5-1))*Xprod. /* 1/(5-1) is 1/(N-1), the inverse of df */
compute invS=inv(S).
print S /title="Covariance Matrix S".
print invS /title="Inverse of S".

*** This section computes the correlation matrix of X1 and X2.
compute rtdiagS=sqrt(diag(S)). /* take the square root of the diagonal elements of S for standard deviations*/
compute rtd=mdiag(rtdiagS). /* mdiag gives the full matrix diagonal with 0 off diagonal elements */
compute invrtd=inv(rtd). /* invrtd is D raised to the negative one-half giving us the reciprocal of the
standard deviations*/
compute R=invrtd*S*invrtd.
print invrtd /title="invrtd: D raised to the negative one-half".
print R /title="R, the correlation matrix".

***This section computes a simple regression with Y regressed on X. The first column of the X matrix must have
ones to obtain the intercept.
compute X={1,-2;
           1,0;
           1,2;
           1,4;
           1,-4}.
compute Y={-1;1;3;-2;-1}.
compute Xtrans=transpos(X).
compute XtransX=Xtrans*X.
compute invXX=inv(XtransX).
compute XtransY=Xtrans*Y.
compute b=invXX*XtransY.
print b /title="b".
end matrix.

```

Output

```

mean of X1 and X2
.00 .00

```

```

Covariance Matrix S
10 1
1 4

```

```

Inverse of S
.1025641026 -.0256410256
-.0256410256 .2564102564

```

```

invrtd: D raised to the negative one-half
.3162277660 .0000000000
.0000000000 .5000000000

```

```

R, the correlation matrix
1.0000000000 .158113883
.158113883 1.0000000000

```

(The first element in the B vector below is the intercept. Here it is zero because I used deviations scores for my X values.)

```

B
.0000000000
.1000000000

```